Supporting Information

Backbone Assignment of Perdeuterated Proteins by Solid-State NMR Using Proton-Detection and Ultrafast Magic-Angle Spinning

Pascal Fricke[1], Veniamin Chevelkov[1*], Maximilian Zinke[1], Karin Giller[2], Stefan Becker[2], Adam Lange[1,3*]

[1]Department of Molecular Biophysics, Leibniz-Institut für Molekulare Pharmakologie, 13125 Berlin, Germany

[2]Department of NMR-based Structural Biology, Max Planck Institute for Biophysical Chemistry, 37077 Göttingen, Germany

[3]Institut für Biologie, Humboldt-Universität zu Berlin, 10115 Berlin, Germany

[*]alange@fmp-berlin.de (A.L.) or shevelkov@fmp-berlin.de (V.S.)

Supplementary Data 1

(H)NH pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; hNH2D_sp
; water suppression by water saturation and spoil pulse
; 2D version
; o1 on water resonance
; o2 in the middle of amide region
; o3 in the middle of carbon region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p23=1.7*p22"
"d31=1s/cnst31-.5*p29-8u-de"
"d0=.5u"

"in0=inf1/4"                                ;d0=F1 -> ph17 (reverse)



1   ze
    d18
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    20u reset:f1  reset:f2 reset:f3


```

```
    20u fq=cnst10:f1                        ;1H offset to HN
    5u pl6:f3                               ;13C XY DC power
    15u pl2:f2                              ;15N hN CP power
    5u  pl3:f1                              ;1H 90 power
    p4:f1 ph1                               ;1H 90 pulse
    .6u pl5:f1                              ;1H hN CP power

     (p15 ph2):f2 (p15:spf3 ph10):f1        ;hN CP


    .6u pl4:f2                              ;15N 90 power
    .6u pl12:f1                             ;1H waltz16 DC power
    .6u cpd1:f1                             ;1H waltz16 DC scheme (pcpd1)
    .6u cpd3:f3                             ;13C xy DC scheme (p28, d28)
    d0                                      ;15N incrementation
    ;(p6 ph21):f2                           ;15N 180 for 1H waltz16 DC optimization
    d0                                      ;15N incrementation
    .6u do:f3                               ;13C DC power OFF
    .6u do:f1                               ;1H DC power OFF

    p5:f2 ph17                              ;15N 90 pulse
    d20                                     ;water suppression delay d20
    2u fq=cnst19:f1                         ;1H offset to water for presat pulse

    1u pl21:f1                              ;1H spoil pulse power
    (p21:spf12 ph13):f1                     ;1H spoil pulse

    d7                                      ;water suppression delay d7

    2u pl16:f1                              ;1H water saturation pulse power
    16  p22:f1 ph22                         ;water saturation
        p23:f1 ph22                         ;water saturation
        p23:f1 ph23                         ;water saturation
    lo to 16 times 1
    d7                                      ;water suppression delay d7

    5u pl5:f1                               ;1H nH CP power
    d9                                      ;delay
    5u fq=cnst10:f1                         ;1H offset to HN
    5u
    p7:f2 ph18                              ;15N 90 pulse
    10u pl2:f2                              ;15N nH CP power

    (p16 ph12):f2  (p16:spf4 ph11):f1       ;nH CP

    5u
    8u pl3:f1                               ;1H power
    d31                                     ;1H refocusing delay
    p29:f1 ph29                             ;1H 180 refocusing pulse
    d31                                     ;1H refocusing delay
    6u pl29:f2                              ;15N waltz16 DC power
    1u  cpd2:f2                             ;15N waltz16 DC scheme (pcpd2)
    1u
    go=2 ph31 ;                             ;Acquisition

    40u do:f1                               ;decouplers off
    40u do:f3                               ;decouplers off
    40u do:f2                               ;decouplers off

    10m wr #0  if #0 zd

    1m ip17
    10u id0
    lo to 2 times td1

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit
```

```
ph1=  0 2
ph10= 3 3
ph2=  0 0 2 2
ph17= 1
ph18= 1 1 1 1 3 3 3 3
ph12= 0
ph13= 3 3 3 3  3 3 3 3  3 3 3 3  3 3 3 3
      1 1 1 1  1 1 1 1  1 1 1 1  1 1 1 1
      3 3 3 3  3 3 3 3  3 3 3 3  3 3 3 3
      1 1 1 1  1 1 1 1  1 1 1 1  1 1 1 1
ph11= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph21= 2
ph29= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph31= 0 2 2 0  2 0 0 2  2 0 0 2  0 2 2 0
ph22= 0 0 0 0  2 2 2 2
ph23= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3

;cnst31: MAS speed in Hz
;d1: recycle delay
;p4: 1H 90° hard excitation pulse duration
;pl3: 1H hard 90° excitation pulse r.f. power
;p15: 1H-15N CP duration
;pl5: 1H CP r.f. power during 1H-15N CP and 15N-1H CP
;pl2: 15N CP r.f. power during 1H-15N CP
;spnam3: ramp shape (on 1H) during 1H-15N CP
;cnst10: distance between water and amide protons in Hz (positive)
;pl12: 1H WALTZ-dec r.f. power during 15N evolution
;pcpd1: 1H WALTZ-dec pulse duration during 15N evolution
;cpdprg1: 1H WALTZ-16 decoupling scheme during 15N evolution
;pl6: 13C XY-dec r.f. power during 15N evolution
;p28: 13C XY-dec pulse duration during 15N evolution
;d28: 13C XY-dec delay during 15N evolution
;cpdprg3: 13C XY decoupling scheme during 15N evolution
;p5: 15N 90° flip pulse duration
;p7: 15N 90° flip pulse duration (for optimization)
;pl4: 15N 90° flip pulse r.f. power
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p21: water suppression spoil pulse duration
;pl21: water suppression spoil pulse r.f. power
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
;p16 15N-1H CP duration
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl29: 15N WALTZ-dec r.f. power during acquisition
;pcpd2: 15N WALTZ-dec pulse duration during acquisition
;cpdprg2: 15N WALTZ-16 decoupling scheme during acquisition
```

Supplementary Data 2

XY decoupling pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; XY_Decoupling
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

1   d28
    p28:0
    d28
    p28:90

    d28
    p28:0
    d28
    p28:90

    d28
    p28:90
    d28
    p28:0

    d28
    p28:90
    d28
    p28:0

    d28
    p28:180
    d28
    p28:270

    d28
    p28:180
    d28
    p28:270

    d28
    p28:270
    d28
    p28:180

    d28
    p28:270
    d28
    p28:180

jump to 1
```

Supplementary Data 3

(H)CANH pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; (H)CANH_3D_sp
; water suppression by water saturation and spoil pulse
; 3D version
; o1 on water resonance
; o2 in the middle of whole carbon region
; o3 in the middle of amide region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d31=1s/cnst31-.5*p29-8u-de"
"p23=1.7*p22"
"d0=.5u"
"d10=.5u"

"in0=inf2/2"                                ;d0=F2 -> ph15 (reverse)
"in10=inf1/4"                               ;d10=F1-> ph16 (reverse)



1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    20u reset:f1  reset:f2 reset:f3



    5u pl29:f3                              ;15N waltz16 DC power
    15u pl2:f2                              ;13C hCA CP power
    5u  pl3:f1                              ;1H 90 power
    2u fq=cnst10:f1                         ;1H offset to HN
    2u fq=cnst11:f2                         ;13C offset to CA
    p4:f1 ph1                               ;1H 90 pulse
    0.6u pl5:f1                             ;1H hCA CP power

    (p15 ph2):f2 (p15:spf1 ph10):f1         ;hCA CP

    .6u  cpd3:f3                            ;15N waltz16 DC scheme
    d0                                      ;13C incrementation
    .6u do:f3                               ;15N DC power OFF

    .6u pl14:f3                             ;15N caN CP power
    .6u pl15:f2                             ;13C caN CP power

    (p25 ph15):f2 (p25:spf3 ph14):f3        ;caN CP

    1u fq=0:f2                              ;13C offset reset
    .6u pl9:f2                              ;13C xy DC power
    .6u cpd5:f2                             ;13C xy DC scheme
    .6u pl12:f1                             ;1H waltz16 DC power
    .6u cpd1:f1                             ;1H waltz16 DC scheme
    d10                                     ;15N incrementation
```

```
    d10                                     ;15N incrementation
    .6u pl6:f3                              ;15N 90 power
    .6u do:f2                               ;13C DC power OFF
    .6u do:f1                               ;1H DC power OFF

    p5:f3 ph16                              ;15N 90 pulse
    d20                                     ;water suppression delay d20
    2u fq=cnst19:f1                         ;1H offset to water for presat pulse

    1u pl21:f1                              ;1H spoil pulse power
    (p21:spf12 ph18):f1                     ;1H spoil pulse

    d7                                      ;water suppression delay d7

    2u pl16:f1                              ;1H water saturation pulse power

    16  p22:f1 ph22                        ;water saturation
        p23:f1 ph23                         ;water saturation
        p23:f1 ph23                         ;water saturation
    lo to 16 times 1

    d7                                      ;water suppression delay d7

    5u pl7:f1                               ;1H nH CP power
    d9                                      ;delay
    5u fq=cnst10:f1                         ;1H offset back to HN
    2u
    p5:f3 ph17                              ;15N 90 pulse
    .6u pl8:f3                              ;15N nH CP power

    (p20 ph12):f3  (p20:spf4 ph11):f1       ;nH CP

    5u
    8u pl3:f1                               ;1H power
    d31                                     ;1H refocusing delay
    p29:f1 ph29                             ;1H 180 refocusing pulse
    d31                                     ;1H refocusing delay

    6u pl30:f3                              ;15N waltz16 DC power
    1u  cpd4:f3                             ;15N waltz16 DC scheme
    1u
    go=2 ph31 ;                             ;Acquisition

    40u do:f1                               ;decouplers off
    40u do:f3                               ;decouplers off
    40u do:f2                               ;decouplers off

    10m wr #0  if #0 zd

    1m ip15
    10u id0
lo to 2 times td2

    800u rp15
    800u rd0

    1m ip16
    800u id10
lo to 2 times td1


HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit

ph1 = 0 2
ph10= 3 3 3 3  3 3 3 3  1 1 1 1  1 1 1 1
ph2 = 0 0 2 2
```

```
ph15= 0
ph14= 0
ph16= 1
ph17= 0 0 0 0  0 0 0 0  0 0 0 0  0 0 0 0
      2 2 2 2  2 2 2 2  2 2 2 2  2 2 2 2
ph18= 0 0 0 0  2 2 2 2
ph12= 1
ph11= 0 0 0 0  2 2 2 2
ph22= 0 0 0 0  2 2 2 2
ph23= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph21= 0
ph29= 0 2 2 0  0 2 2 0
ph31= 0 2 2 0  2 0 0 2   2 0 0 2  0 2 2 0
      2 0 0 2  0 2 2 0   0 2 2 0  2 0 0 2

;cnst31: MAS speed in Hz
;d1: recycle delay
;p4: 1H 90° hard excitation pulse duration
;pl3: 1H hard 90° excitation pulse r.f. power
;cnst10: distance between water and amide protons in Hz (positive)
;cnst11: distance between 13C offset and 13CA in Hz (negative)
;p15: 1H-13CA CP duration
;pl5: 1H CP r.f. power during 1H-13CA CP
;pl2: 13CA CP r.f. power during 1H-13CA CP
;spnam1: ramp shape (on 1H) during 1H-13CA CP
;pl29: 15N WALTZ-dec r.f. power during 13CA evolution
;pcpd3: 15N WALTZ-dec pulse duration during 13CA evolution and acquisition
;cpdprg3: 15N WALTZ-16 decoupling scheme during 13CA evolution
;p25: 13CA-15N CP duration
;pl15: 13CA CP r.f. power during 13CA-15N CP
;pl14: 15N CP r.f. power during 13CA-15N CP
;spnam3: ramp shape (on 15N) during 13CA-15N CP
;pl9: 13C XY-dec r.f. power during 15N evolution
;p28: 13C XY-dec pulse duration during 15N evolution
;d28: 13C XY-dec delay during 15N evolution
;cpdprg5: 13C XY decoupling scheme during 15N evolution
;pl12: 1H WALTZ-dec r.f. power during 15N evolution
;pcpd1: 1H WALTZ-dec pulse duration during 15N evolution
;cpdprg1: 1H WALTZ-16 decoupling scheme during 15N evolution
;p5: 15N 90° flip pulse duration
;pl6: 15N 90° flip pulse r.f. power
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p21: water suppression spoil pulse duration
;pl21: water suppression spoil pulse r.f. power
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
;p20: 15N-1H CP duration
;pl8: 15N CP r.f. power during 15N-1H CP
;pl7: 1H CP r.f. power during 15N-1H CP
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl30: 15N WALTZ-dec r.f. power during acquisition
;cpdprg4: 15N WALTZ-16 decoupling scheme during acquisition
```

Supplementary Data 4

(H)CONH pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; (H)CONH_3D
; water supression by water saturation and spoil pulse
; 3D version
; o1 on water resonance
; o2 in the middle of CO
; o3 in the middle of amide region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d31=1s/cnst31-.5*p29-8u-de"
"p23=p22*1.7"
"d0=0.5u"
"d10=.5u"

"in0=inf2/4"                                    ;d0= F2 -> ph15
"in10=inf1/4"                                   ;d10=F1 -> ph16  (reverse)


1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1

    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    20u reset:f1  reset:f2 reset:f3



    15u pl2:f2                                  ;13C hCO CP power
    5u  pl3:f1                                  ;1H 90 power
    2u fq=cnst10:f1                             ;1H offset to HN
    1u fq=0:f2                                  ;13C offset reset
    p4:f1 ph1                                   ;1H 90 pulse
    0.4u pl5:f1                                 ;1H hCO CP power

    (p15 ph2):f2 (p15:spf1 ph10):f1             ;hCO CP

    10u
    2u pl6:f3                                   ;15N power for refoc pulses
    d0                                          ;13C incrementation
    .6u pl26:f2                                 ;13C selective pulse power
    0.1u
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13CA selective pulse + 15N 180 refoc pulse
    d0                                          ;13C incrementation
    10u
    (p26:sp6 ph25):f2                           ;13CO selective pulse (spoffs6, sp6, spnam6)
    10u
    2u  pl12:f1                                 ;1H waltz16 DC power
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13CA selective pulse + 15N 180 refoc pulse

    10u  pl14:f3                                ;15N coN CP power
    2u pl15:f2                                  ;13C coN CP power

```

```
    (p25 ph15):f2 (p25:spf3 ph14):f3          ;coN CP

    .6u pl9:f2                                ;13C xy DC power
    1u fq=cnst24:f2                           ;13C offset to center
    .6u cpd5:f2                               ;13C xy DC scheme
    .8u cpd1:f1                               ;1H waltz16 DC scheme

    d10                                       ;15N incrementation
    d10                                       ;15N incrementation
    .6u pl6:f3                                ;15N 90 power
    .6u do:f2                                 ;13C DC power OFF
    .6u do:f1                                 ;1H DC power OFF

    p5:f3 ph16                                ;15N 90 pulse
    d20                                       ;water suppression delay d20
    2u fq=cnst19:f1                           ;1H offset to water

    1u pl21:f1                                ;1H spoil pulse power
     (p21:spf12 ph7):f1                       ;1H spoil pulse

    d7                                        ;water suppression delay d7

    2u pl16:f1                                ;1H water saturation pulse power
    16  p22:f1 ph22                           ;water saturation
        p23:f1 ph23                           ;water saturation
        p23:f1 ph23                           ;water saturation
    lo to 16 times 1

    d7                                        ;water suppression delay d7

    5u pl7:f1                                 ;1H nH CP power
    d9                                        ;delay
    5u fq=cnst10:f1                           ;1H offset back to HN
    2u
    p5:f3 ph17                                ;15N 90 pulse
    .6u pl8:f3                                ;15N nH CP power

    (p20 ph12):f3  (p20:spf4 ph11):f1         ;nH CP

    5u
    8u pl3:f1                                 ;1H power
    d31                                       ;1H refocusing delay
    p29:f1 ph29                               ;1H 180 refocusing pulse
    d31                                       ;1H refocusing delay

    6u pl30:f3                                ;15N waltz16 DC power
    1u  cpd4:f3                               ;15N waltz16 DC scheme
    1u
    go=2 ph31                                 ;Acquisition
    40u do:f1                                 ;decouplers off
    40u do:f3                                 ;decouplers off
    40u do:f2                                 ;decouplers off

    10m wr #0  if #0 zd

    1m ip15
    10u id0
lo to 2 times td2

    800u rp15
    800u rd0

    1m ip16
    800u id10
lo to 2 times td1

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF
```

```
exit

ph1=  0 2
ph10= 3 3 1 1
ph2=  0 0 0 0  2 2 2 2
ph7 = 1 1 1 1  3 3 3 3
ph15= 0
ph14= 0
ph16= 1
ph17= 0 0 0 0  0 0 0 0  0 0 0 0  0 0 0 0
      2 2 2 2  2 2 2 2  2 2 2 2  2 2 2 2
ph12= 1
ph11= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph22= 2 0 0 2  0 2 2 0  1 3 3 1  3 1 1 3
ph23= 1 1 3 3  3 3 1 1  0 0 2 2  2 2 0 0
ph29= 0 2 2 0  0 2 2 0  2 0 0 2  2 0 0 2
ph25= 0
ph31= 0 2 2 0  2 0 0 2  2 0 0 2  0 2 2 0
      2 0 0 2  0 2 2 0  0 2 2 0  2 0 0 2

;cnst31: MAS speed in Hz
;d1: recycle delay
;p4: 1H 90° hard excitation pulse duration
;pl3: 1H hard 90° excitation pulse r.f. power
;cnst10: distance between water and amide protons in Hz (positive)
;p15: 1H-13CO CP duration
;pl5: 1H CP r.f. power during 1H-13CO CP
;pl2: 13CO CP r.f. power during 1H-13CO CP
;spnam1: ramp shape (on 1H) during 1H-13CO CP
;pl6: 15N hard pulse r.f. power
;p26: 13CA/CO selective pulse duration
;pl26: 13CA/CO selective pulse r.f. power
;sp5: 13CA selective pulse r.f. power
;sp6: 13CO selective pulse r.f. power
;spnam5: 13CA selective pulse shape
;spnam6: 13CO selective pulse shape
;spoffs5: distance between 13CO and 13CA (negative)
;spoffs6: 0 Hz
;p11: 15N 180° inversion pulse duration
;p25: 13CO-15N CP duration
;pl15: 13CO CP r.f. power during 13CO-15N CP
;pl14: 15N CP r.f. power during 13CO-15N CP
;spnam3: ramp shape (on 15N) during 13CO-15N CP
;cnst24: distance between CO and center (100ppm) in Hz (negative)
;pl9: 13C XY-dec r.f. power during 15N evolution
;p28: 13C XY-dec pulse duration during 15N evolution
;d28: 13C XY-dec delay during 15N evolution
;cpdprg5: 13C XY decoupling scheme during 15N evolution
;pl12: 1H WALTZ-dec r.f. power during 15N evolution
;pcpd1: 1H WALTZ-dec pulse duration during 15N evolution
;cpdprg1: 1H WALTZ-16 decoupling scheme during 15N evolution
;p5: 15N 90° flip pulse duration
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p21: water suppression spoil pulse duration
;pl21: water suppression spoil pulse r.f. power
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
;p20: 15N-1H CP duration
;pl8: 15N CP r.f. power during 15N-1H CP
;pl7: 1H CP r.f. power during 15N-1H CP
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl30: 15N WALTZ-dec r.f. power during acquisition
;cpdprg4: 15N WALTZ-16 decoupling scheme during acquisition
```

Supplementary Data 5

Selective 180° ^{13}C pulse optimization sequence example (on Bruker hardware running TopSpin version 2.1)

```
; Selective_180
; selective soft pulse on either CA or CO and subsequent carbon 90 hard pulse
; used for CA decoupling during CO evolution periods
; o1 in the middle of whole carbon region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>



1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f2
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    1u fq=0:f2
    1u pl11:f1                              ;13C selective pulse power
    5u
    (p11:sp11 ph3):f1                       ;13C selective pulse (spoffs11, sp11, spnam11)
    5u
    1u pl1:f1                               ;13C 90 power
    p1:f1 ph1                               ;13C 90 pulse
    1u pl12:f2                              ;1H spinal64 DC power (optional)
    1u cpds2:f2                             ;1H spinal64 DC scheme
    go=2 ph31                               ;Acquisition

    1m do:f2                                ;decouplers off

    100m wr #0

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit

ph1 = 0 1 2 3
ph3 = 0 1 2 3
ph31= 2 3 0 1

;d1: recycle delay
;p11: 13CA/CO selective pulse duration
;pl11: 13CA/CO selective pulse power
;sp11: 13CA/CO selective pulse power
;spnam11: 13CA/CO selective pulse shape
;spoffs11: distance between carbon offset and CA/CO in Hz
;p1: 13C hard excitation pulse duration
;pl1: 13C hard excitation pulse r.f. power
;pl12: 1H spinal 64 decoupling r.f. power (not needed, can be 0)
;cpdprg2: 1H spinal 64 decoupling scheme
```

Supplementary Data 6

(H)CACO(N)H pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; (H)CACO(N)H_3D_sp
; water suppression by water saturation and spoil pulse
; 3D version
; o1 on water resonance
; o2 in the middle of CA
; o3 in the middle of amide region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d31=1s/cnst31-5u-.5*p29-de"
"p23=1.7*p22"
"d0=.5u"
"d10=.5u"

"in0=inf2/2"                                    ;d0=F2 -> ph2
"in10=inf1/4"                                   ;d10=F1-> ph15


1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    1m reset:f1 reset:f2 reset:f3


    1u fq=0:f2                                  ;13C offset reset
    1u fq=cnst10:f1                             ;1H offset to HN
    10u pl2:f2                                  ;13C hCA CP power
    10u  pl1:f1                                 ;1H 90 power
    10u pl29:f3                                 ;15N waltz16 DC power
    p2:f1 ph1                                   ;1H 90 pulse
    .6u pl19:f1                                 ;1H hCA CP power

    (p15:spf1 ph0):f1 (p15 ph2):f2              ;hCA CP

    .7u  cpd4:f3                                ;15N waltz16 DC scheme (pcpd3)
    0.1u
    d0                                          ;13C incrementation
    .6u pl17:f2                                 ;13C caCO homo CP power
    1u do:f3                                    ;15N DC power OFF

    (p17:spf17 ph17):f2                         ;13C caCO homo CP

    .6u fq=cnst25:f2                            ;13C offset to CO
    .6u pl18:f2                                 ;13C trim pulse power
    (p18 ph18):f2                               ;13CO trim pulse
    10u
    2u pl6:f3                                   ;15N power
    d10                                         ;13C incrementation
    .6u pl26:f2                                 ;13C selective pulse power
    0.1u
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13CA selective pulse + 15N 180 refoc pulse
```

```
    d10                                         ;13C incrementation
    10u
    (p26:sp6 ph25):f2                           ;13CO selective pulse (spoffs6, sp6)
    10u
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13Calpha selective pulse + 15N 180 refoc pulse
    10u  pl14:f3                                ;15N coN CP power
    2u pl15:f2                                  ;13C coN CP power

    (p25 ph15):f2 (p25:spf3 ph14):f3            ;coN CP

    .6u pl6:f3                                  ;15N 90 power
    p5:f3 ph16                                  ;15N 90 pulse
    d20                                         ;water suppression delay d20
    2u fq=cnst19:f1                             ;1H offset to water

    1u pl7:f1                                   ;1H spoil pulse power
    (p14:spf12 ph24):f1                         ;1H spoil pulse

    d7                                          ;water suppression delay d7

    2u pl16:f1                                  ;1H water saturation power
    16  p22:f1 ph22                             ;water saturation
        p23:f1 ph23                             ;water saturation
        p23:f1 ph23                             ;water saturation
    lo to 16 times 1

    d7                                          ;water suppression delay d7

    5u pl7:f1                                   ;1H nH CP power
    5u fq=cnst10:f1                             ;1H offset to HN
    d9                                          ;delay
    p5:f3 ph20                                  ;15N 90 pulse
    .6u pl8:f3                                  ;15N nH CP power

    (p16 ph12):f3  (p16:spf4 ph11):f1           ;nH CP

    5u
    5u pl1:f1                                   ;1H power
    d31                                         ;1H refoc delay
    p29:f1 ph29                                 ;1H 180 refoc pulse
    d31                                         ;1H refoc delay

    6u pl30:f3                                  ;15N waltz16 DC power
    1u  cpd6:f3                                 ;15N waltz16 DC scheme (pcpd3)
    1u
    go=2 ph31                                   ;Acquisition
    40u do:f1                                   ;decouplers off
    40u do:f3                                   ;decouplers off
    40u do:f2                                   ;decouplers off

    10m wr #0  if #0 zd

    1m ip2
    10u id0
lo to 2 times td2

    500u rp2
    100u rd0

    600u ip15
    200u id10
lo to 2 times td1

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit

ph0=  3 3 3 3  3 3 3 3  1 1 1 1  1 1 1 1
```

```
ph1=  0 2
ph2=  0 0 2 2
ph17= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph18= 3 3 3 3  3 3 3 3  1 1 1 1  1 1 1 1
ph14= 1 1 1 1  1 1 1 1  1 1 1 1  1 1 1 1
      3 3 3 3  3 3 3 3  3 3 3 3  3 3 3 3
ph15= 0
ph16= 2
ph20= 1
ph12= 0
ph22= 0 0 0 0  2 2 2 2
ph23= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph11= 0 0 0 0  2 2 2 2
ph24= 0 0 0 0  2 2 2 2
ph29= 0 0 2 2
ph25= 0
ph31= 0 2 2 0  2 0 0 2  2 0 0 2  0 2 2 0
      2 0 0 2  0 2 2 0  0 2 2 0  2 0 0 2

;cnst31: MAS speed in Hz
;d1: recycle delay
;p2: 1H 90° hard excitation pulse duration
;pl1: 1H hard pulse r.f. power
;cnst10: distance between water and amide protons in Hz (positive)
;p15: 1H-13CA CP duration
;pl19: 1H CP r.f. power during 1H-13CA CP
;pl2: 13CA CP r.f. power during 1H-13CA CP
;spnam1: ramp shape (on 1H) during 1H-13CA CP
;pl29: 15N WALTZ-dec r.f. power during 13CA evolution
;pcpd3: 15N WALTZ-dec pulse duration during 13CA evolution
;cpdprg4: 15N WALTZ-16 decoupling scheme during 13CA evolution
;p17: 13CA-13CO DREAM pulse duration
;pl17: 13CA-13CO DREAM pulse r.f. power
;spnam17: 13CA-13CO DREAM pulse shape
;cnst25: distance between 13CA and 13CO in Hz (positive)
;p18: trim pulse duration
;pl18: trim pulse r.f. power
;p26: 13CA/CO selective pulse duration
;pl26: 13CA/CO selective pulse r.f. power
;sp5: 13CA selective pulse r.f. power
;sp6: 13CO selective pulse r.f. power
;spnam5: 13CA selective pulse shape
;spnam6: 13CO selective pulse shape
;spoffs5: distance between 13CO and 13CA (negative)
;spoffs6: 0 Hz
;p11: 15N 180° inversion pulse duration
;pl6: 15N hard pulse r.f. power
;p25: 13CO-15N CP duration
;pl15: 13CO CP r.f. power during 13CO-15N CP
;pl14: 15N CP r.f. power during 13CO-15N CP
;spnam3: ramp shape (on 15N) during 13CO-15N CP
;p5: 15N 90° flip pulse duration
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p14: water suppression spoil pulse duration
;pl7: water suppression spoil pulse r.f. power and 1H CP r.f. power for 15N-1H CP
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
;p16: 15N-1H CP duration
;pl8: 15N CP r.f. power during 15N-1H CP
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl30: 15N WALTZ-dec r.f. power during acquisition
;cpdprg6: 15N WALTZ-16 decoupling scheme during acquisition
;pcpd3: 15N WALTZ decoupling pulse duration
```

Supplementary Data 7

^{13}CA-^{13}CO or ^{13}CA-^{13}CB DREAM pulse optimization sequence example (on Bruker hardware running TopSpin version 2.1)

```
; CACO/CACB_DREAM_calib
; SPECIFIC CP on CA and subsequent DREAM transfer to CO or to CB
; used for CACO or CACB transfer
; trim pulse only needed for CACO DREAM
; comment line 42 for transfer from CA to CB!
; o1 in the middle of whole carbon region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>



1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   10u do:f2
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    10u fq=0:f2
    10u
    10u fq=cnst15:f1                        ;13C offset to CA
    10u pl2:f2                              ;1H 90 power
    5u pl5:f1                               ;13C hCA CP power
    p2:f2 ph1                               ;1H 90 pulse
    0.8u pl6:f2                             ;1H hCA CP power

    (p15 ph2):f1 (p15:spf0 pl6 ph0):f2      ;hCA CP

    0.8u pl17:f1                            ;13C caCO or caCB homo CP power

    (p17:spf17 ph17):f1                     ;13C caCO or caCB homo CP

    1u pl18:f1                              ;13CO trim pulse power
    .8u fq=cnst25:f1                        ;13C offset to CO
    (p18 ph18):f1                           ;13CO trim pulse (comment line for CACB DREAM!)
    .8u fq=0:f1                             ;13C offset reset

    1u pl12:f2                              ;1H spinal64 DC power (optional)
    1u cpds2:f2                             ;1H spinal64 DC scheme
    go=2 ph31                               ;Acquisition

    1m do:f2                                ;decouplers off

    100m  wr #0

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit


ph0= 0 2
ph1= 1 1 1 1  3 3 3 3
ph20=3 3 3 3  1 1 1 1
ph2= 0 0 2 2
ph3= 1 1 3 3
```

```
ph17=0
ph18=3
ph31=0 2 2 0  2 0 0 2

;d1: recycle delay
;cnst15: distance between carbon offset and 13CA in Hz (negative)
;p2: 1H 90° excitation pulse duration
;pl2: 1H 90° excitation pulse r.f. power
;p15: 1H-13CA CP duration
;pl5: 13CA CP r.f. power during 1H-13CA CP
;pl6: 1H CP r.f. power during 1H-13CA CP
;spnam0: ramp shape (on 1H) during 1H-13CA CP
;p17: 13CA-13CO or 13CA-13CB DREAM pulse duration
;pl17: 13CA-13CO or 13CA-13CB DREAM pulse r.f. power
;spnam17: 13CA-13CO or 13CA-13CB DREAM pulse shape
;p18: trim pulse duration
;pl18: trim pulse r.f. power
;cnst25: distance between carbon offset and 13CO in Hz (positive)
;pl12: 1H spinal 64 decoupling r.f. power (not needed, can be 0)
;cpdprg2: 1H spinal 64 decoupling scheme
```

Supplementary Data 8

(H)COCA(N)H pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; (H)COCA(N)H_3D_sp
; water suppression by water saturation and spoil pulse
; 3D version
; o1 on water resonance
; o2 in the middle of CO
; o3 in the middle of amide region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d31=1s/cnst31-5u-.5*p29-de"
"p23=1.7*p22"
"d0=.5u"
"d10=.5u"

"in0=inf2/4"                                    ;d0=F2 -> ph15 (reverse)
"in10=inf1/4"                                   ;d10=F1-> ph2


1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    1m reset:f1 reset:f2 reset:f3


    1u fq=0:f2                                  ;13C offset reset
    1u fq=cnst10:f1                             ;1H offset to HN
    10u pl2:f2                                  ;13C hCO CP power
    10u  pl1:f1                                 ;1H 90 power
    p2:f1 ph1                                   ;1H 90 pulse
    .6u pl19:f1                                 ;1H hCO CP power

    (p15:spf1 ph0):f1 (p15 ph2):f2              ;hCO CP

    10u
    2u pl6:f3                                   ;15N power for refoc pulses
    d10                                         ;13C incrementation
    .6u pl26:f2                                 ;13C selective pulse power
    0.1u
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13CA selective pulse + 15N 180 refoc pulse
    d10                                         ;13C incrementation
    10u
    (p26:sp6 ph25):f2                           ;13CO selective pulse (spoffs6, sp6, spnam6)
    10u
    (center (p26:sp5 ph25):f2  (p11 ph25):f3)   ;13CA selective pulse + 15N 180 refoc pulse

    2u  pl29:f3                                 ;15N waltz16 DC power
    .6u pl18:f2                                 ;13C trim pulse power
    (p18 ph18):f2                               ;13CO trim pulse
    .6u fq=cnst0:f2                             ;13C offset to CA
    .6u pl17:f2                                 ;13C coCA homo CP power

```

```
    (p17:spf17 ph17):f2                         ;coCA homo CP

    .6u  cpd4:f3                                ;15N waltz16 DC scheme (pcpd3)
    d0                                          ;13C incrementation
    d0                                          ;13C incrementation
    1u do:f3                                    ;15N DC power OFF
    2u pl15:f2                                  ;13C caN CP power
    2u pl14:f3                                  ;15N caN CP power
    2u

    (p25 ph15):f2 (p25:spf3 ph14):f3            ;caN CP

    2u pl6:f3                                   ;15N 90 power
    p5:f3 ph16                                  ;15N 90 pulse
    d20                                         ;water suppression delay d20
    2u fq=cnst19:f1                             ;1H offset to water

    1u pl7:f1                                   ;1H spoil pulse power
    (p14:spf12 ph24):f1                         ;1H spoil pulse

    d7                                          ;water suppression delay d7

    2u pl16:f1                                  ;1H water saturation power
    16  p22:f1 ph22                             ;water saturation
        p23:f1 ph23                             ;water saturation
        p23:f1 ph23                             ;water saturation
    lo to 16 times 1

    d7                                          ;water suppression delay d7

    5u pl7:f1                                   ;1H nH CP power
    d9                                          ;delay
    5u fq=cnst10:f1                             ;1H offset to HN
    2u
    p5:f3 ph20                                  ;15N 90 pulse
    .6u pl8:f3                                  ;15N nH CP power

    (p16 ph12):f3  (p16:spf4 ph11):f1           ;nH CP

    5u
    5u pl1:f1                                   ;1H power
    d31                                         ;1H refoc delay
    p29:f1 ph29                                 ;1H 180 refoc pulse
    d31                                         ;1H refoc delay

    6u pl30:f3                                  ;15N waltz16 DC power
    1u cpd6:f3                                  ;15N waltz16 DC scheme  (pcpd3)
    1u
    go=2 ph31                                   ;Acquisition
    40u do:f1                                   ;decouplers off
    40u do:f3                                   ;decouplers off
    40u do:f2                                   ;decouplers off

    10m wr #0  if #0 zd

    1m ip15
    10u id0
lo to 2 times td2

    100u rp15
    10u rd0

    100u ip2
    100u id10
lo to 2 times td1

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF
```

```
exit

ph0=  0 2
ph1=  1 1 1 1  3 3 3 3
ph2=  0 0 2 2
ph17= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph18= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph14= 1
ph15= 0
ph16= 2
ph20= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph12= 0
ph22= 0 0 0 0  2 2 2 2
ph23= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph24= 0 0 0 0  2 2 2 2
ph11= 0 0 0 0  0 0 0 0  1 1 1 1  1 1 1 1
      2 2 2 2  2 2 2 2  3 3 3 3  3 3 3 3
      1 1 1 1  1 1 1 1  0 0 0 0  0 0 0 0
      3 3 3 3  3 3 3 3  2 2 2 2  2 2 2 2
ph29= 0 0 0 0  0 0 0 0  1 1 1 1  1 1 1 1
      2 2 2 2  2 2 2 2  3 3 3 3  3 3 3 3
      1 1 1 1  1 1 1 1  0 0 0 0  0 0 0 0
      3 3 3 3  3 3 3 3  2 2 2 2  2 2 2 2
ph25= 0
ph31= 0 2 2 0  2 0 0 2  3 1 1 3  1 3 3 1
      2 0 0 2  0 2 2 0  1 3 3 1  3 1 1 3
      1 3 3 1  3 1 1 3  2 0 0 2  0 2 2 0
      3 1 1 3  1 3 3 1  0 2 2 0  2 0 0 2

;cnst31: MAS speed in Hz
;d1: recycle delay
;p2: 1H 90° hard excitation pulse duration
;pl1: 1H hard pulse r.f. power
;cnst10: distance between water and amide protons in Hz (positive)
;p15: 1H-13CO CP duration
;pl19: 1H CP r.f. power during 1H-13CO CP
;pl2: 13CO CP r.f. power during 1H-13CO CP
;spnam1: ramp shape (on 1H) during 1H-13CO CP
;p26: 13CA/CO selective pulse duration
;pl26: 13CA/CO selective pulse r.f. power
;sp5: 13CA selective pulse r.f. power
;sp6: 13CO selective pulse r.f. power
;spnam5: 13CA selective pulse shape
;spnam6: 13CO selective pulse shape
;spoffs5: distance between 13CO and 13CA (negative)
;spoffs6: 0 Hz
;p11: 15N 180° inversion pulse duration
;pl6: 15N hard pulse r.f. power
;p18: trim pulse duration
;pl18: trim pulse r.f. power
;cnst0: distance between 13CO and 13CA in Hz (negative)
;p17: 13CO-13CA DREAM pulse duration
;pl17: 13CO-13CA DREAM pulse r.f. power
;spnam17: 13CO-13CA DREAM pulse shape
;pl29: 15N WALTZ-dec r.f. power during 13CA evolution
;pcpd3: 15N WALTZ-dec pulse duration during 13CA evolution
;cpdprg4: 15N WALTZ-16 decoupling scheme during 13CA evolution
;p25: 13CA-15N CP duration
;pl15: 13CA CP r.f. power during 13CA-15N CP
;pl14: 15N CP r.f. power during 13CA-15N CP
;spnam3: ramp shape (on 15N) during 13CA-15N CP
;p5: 15N 90° flip pulse duration
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p14: water suppression spoil pulse duration
;pl7: water suppression spoil pulse r.f. power and 1H CP r.f. power for 15N-1H CP
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
```

```
;p16: 15N-1H CP duration
;pl8: 15N CP r.f. power during 15N-1H CP
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl30: 15N WALTZ-dec r.f. power during acquisition
;cpdprg6: 15N WALTZ-16 decoupling scheme during acquisition
;pcpd3: 15N WALTZ decoupling pulse duration
```

Supplementary Data 9

(H)CBCA(N)H pulse sequence example (on Bruker hardware running TopSpin version 2.1)

```
; (H)CBCA(N)H_3D_sp
; water suppression by water saturation and spoil pulse
; 3D version
; o1 on water resonance
; o2 in the middle of CB
; o3 in the middle of amide region
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"d31=1s/cnst31-5u-.5*p29-de"
"p23=1.7*p22"
"d0=0.5u"
"d10=0.5u"

"in0=inf2/4"                                ;d0=F2 -> ph15 (reverse)
"in10=inf1/4"                               ;d10=F1-> ph2



1   ze
    30m LOCKDEC_ON
    30m LOCKH_ON
    30m H2_PULSE
2   40u do:f3
    40u do:f2
    40u do:f1
    30m H2_LOCK
    30m LOCKH_OFF
    d1
    30m LOCKH_ON
    30m H2_PULSE
    1m reset:f1 reset:f2 reset:f3


    1u fq=cnst25:f2                         ;13C offset reset
    1u fq=cnst10:f1                         ;1H offset to HN
    10u pl2:f2                              ;13C hCB CP power
    10u  pl1:f1                             ;1H 90 power
    p2:f1 ph1                               ;1H 90 pulse
    .6u pl19:f1                             ;1H hCB CP power

    (p15:spf1 ph0):f1 (p15 ph2):f2          ;hCB CP

    10u fq=cnst20:f2                        ;13C offset to adjust indirect dimension
    d10                                     ;13CB incrementation
    d10                                     ;13CB incrementation
    .6u  pl29:f3                            ;15N waltz16 DC power
    .6u fq=cnst0:f2                         ;13C offset to CA
    .6u pl17:f2                             ;13C cbCA homo CP power
    (p17:spf17 ph17):f2                     ;cbCA homo CP (spnam17)
    .7u  cpd4:f3                            ;15N waltz16 DC (pcpd3)
    d0                                      ;13CA incrementation
    d0                                      ;13CA incrementation
    1u do:f3                                ;15N DC power OFF
    .8u pl15:f2                             ;13C caN CP power
    2u
    .8u pl14:f3                             ;15N caN CP power

    (p25 ph15):f2 (p25:spf3 ph14):f3        ;caN CP

```

```
    2u do:f1                                ;1H DC power OFF

    .6u pl6:f3                              ;15N 90 power
    p5:f3 ph16                              ;15N 90 pulse
    d20                                     ;water suppression delay d20
    2u fq=cnst19:f1                         ;1H offset to water

    1u pl7:f1                               ;1H spoil pulse power
    (p14:spf12 ph24):f1                     ;1H spoil pulse

    d7                                      ;water suppression delay d7

    2u pl16:f1                              ;1H water saturation power
    16  p22:f1 ph22                         ;water saturation
        p23:f1 ph23                         ;water saturation
        p23:f1 ph23                         ;water saturation
    lo to 16 times 1
    d7                                      ;water suppression delay d7

    5u pl7:f1                               ;1H nH CP power
    5u fq=cnst10:f1                         ;1H offset to HN
    d9                                      ;delay
    p5:f3 ph20                              ;15N 90 pulse
    .6u pl8:f3                              ;15N nH CP power

    (p16 ph12):f3  (p16:spf4 ph11):f1       ;nH CP

    5u
    5u pl1:f1                               ;1H power
    d31                                     ;1H refoc delay
    p29:f1 ph29                             ;1H 180 refoc pulse
    d31                                     ;1H refoc delay
    3u
    1u pl30:f3                              ;15N waltz16 DC power
    1u  cpd6:f3                             ;15N waltz16 DC scheme (pcpd3)
    1u
    go=2 ph31                               ;Acquisition
    40u do:f1                               ;decouplers off
    40u do:f3                               ;decouplers off
    40u do:f2                               ;decouplers off

    10m wr #0  if #0 zd

    1m ip15
    100u id0
lo to 2 times td2

    100u rp15
    100u rd0

    100u ip2
    100u id10
lo to 2 times td1

HaltAcqu, 1m

    30m H2_LOCK
    30m LOCKH_OFF
    30m LOCKDEC_OFF

exit

ph0=  0 2
ph1=  1 1 1 1  3 3 3 3
ph2=  0 0 2 2
ph17= 0 0 0 0  0 0 0 0  2 2 2 2  2 2 2 2
ph14= 1
ph15= 0
ph16= 2
ph20= 1 1 1 1  1 1 1 1  3 3 3 3  3 3 3 3
ph12= 0
ph22= 0 0 0 0  2 2 2 2
```

```
ph23= 1 1 1 1   1 1 1 1   3 3 3 3   3 3 3 3
ph24= 0 0 0 0   2 2 2 2
ph11= 0 0 0 0   0 0 0 0   1 1 1 1   1 1 1 1
      2 2 2 2   2 2 2 2   3 3 3 3   3 3 3 3
      1 1 1 1   1 1 1 1   0 0 0 0   0 0 0 0
      3 3 3 3   3 3 3 3   2 2 2 2   2 2 2 2
ph29= 0 0 0 0   0 0 0 0   1 1 1 1   1 1 1 1
      2 2 2 2   2 2 2 2   3 3 3 3   3 3 3 3
      1 1 1 1   1 1 1 1   0 0 0 0   0 0 0 0
      3 3 3 3   3 3 3 3   2 2 2 2   2 2 2 2
ph31= 0 2 2 0   2 0 0 2   3 1 1 3   1 3 3 1
      2 0 0 2   0 2 2 0   1 3 3 1   3 1 1 3
      1 3 3 1   3 1 1 3   2 0 0 2   0 2 2 0
      3 1 1 3   1 3 3 1   0 2 2 0   2 0 0 2

;cnst31: MAS speed in Hz
;d1: recycle delay
;cnst25: 0
;p2: 1H 90° hard excitation pulse duration
;pl1: 1H hard pulse r.f. power
;cnst10: distance between water and amide protons in Hz (positive)
;p15: 1H-13CB CP duration
;pl19: 1H CP r.f. power during 1H-13CB CP
;pl2: 13CB CP r.f. power during 1H-13CB CP
;spnam1: ramp shape (on 1H) during 1H-13CB CP
;cnst20: adjustment (in Hz) to correctly center 13CB indirect dimension
;cnst0: distance between 13CB and 13CA in Hz (positive)
;p17: 13CB-13CA DREAM pulse duration
;pl17: 13CB-13CA DREAM pulse r.f. power
;spnam17: 13CB-13CA DREAM pulse shape
;pl29: 15N WALTZ-dec r.f. power during 13CA evolution
;pcpd3: 15N WALTZ-dec pulse duration during 13CA evolution
;cpdprg4: 15N WALTZ-16 decoupling scheme during 13CA evolution
;p25: 13CA-15N CP duration
;pl15: 13CA CP r.f. power during 13CA-15N CP
;pl14: 15N CP r.f. power during 13CA-15N CP
;spnam3: ramp shape (on 15N) during 13CA-15N CP
;pl6: 15N 90° flip pulse r.f. power
;p5: 15N 90° flip pulse duration
;d20: water suppression T delay
;cnst19: distance between 1H offset and water signal in Hz
;p14: water suppression spoil pulse duration
;pl7: water suppression spoil pulse r.f. power and 1H CP r.f. power for 15N-1H CP
;spnam12: water suppression spoil pulse ramp shape
;p22: first pulse duration in water suppression train
;p23: second pulse duration in water suppression train (automatically calculated)
;pl16: water suppression train r.f. power
;d7: small water suppression delay to optimize efficiency
;d9: small water suppression delay to optimize efficiency
;p16: 15N-1H CP duration
;pl8: 15N CP r.f. power during 15N-1H CP
;spnam4: ramp shape (on 1H) during 15N-1H CP
;d31: spin echo delay (automatically calculated)
;p29: 1H 180°spin echo pulse duration
;pl30: 15N WALTZ-dec r.f. power during acquisition
;cpdprg6: 15N WALTZ-16 decoupling scheme during acquisition
;pcpd3: 15N WALTZ decoupling pulse duration
```

Supplementary Data 10

Gradient water suppression pulse sequence snippet

```
; Gradient for water suppression
;*******************************************************
;* The authors are not responsible for any damage caused *
;*   by the use of this pulse program or parts thereof.  *
;*******************************************************

    50u UNBLKGRAD
    p27:gp21                                ;gradient application (gpnam21, gpz21, p17, d27)
    d27                                     ;ringdown delay
    5u BLKGRAD
```